Chapter 10
Energy System Integration

This chapter provides a solution for energy system integration (ESI) problem in GAMS. The ESI analysis refers to a class of studies which investigate the potential in different energy sectors (water, gas, and electricity) for moving toward a more environmentally friendly and efficient energy supply. The main idea is how to harvest the flexibilities in each energy carrier in a larger framework. In this chapter, the coordination between water desalination systems and power system, gas network-power network, and finally the concept of energy hub is investigated.

10.1 Water-Power Nexus

The water-electricity interdependence is an undeniable issue. Water is used for cooling the power plants. On the other hand, the electricity is used for collecting, treatment, and disposal of water.

The water-energy nexus concept is shown in Fig. 10.1 [1]. The optimization is formulated as follows:

$$\min_{\text{DV}} \text{OF} = \text{TC} + \text{CC} + \text{WC} \tag{10.1a}$$

$$\text{DV} = \left\{P_{g,t}, P_{c,t}, W_{c,t}, W_{w,t}\right\} \tag{10.1b}$$

$$\text{CT} = \sum_{g,t} a_g P_{g,t}^2 + b_g P_{g,t} + c_g U_{g,t} \tag{10.1c}$$

$$\text{CC} = \sum_{c,t} \alpha_c P_{c,t}^2 + \beta_c P_{c,t} W_{c,t} + \gamma_c W_{c,t}^2 + \zeta_c P_{c,t} + \varsigma_c W_{c,t} + \xi_c U_{c,t} \tag{10.1d}$$

$$\text{CW} = \sum_{w,t} a_w W_{w,t}^2 + b_w W_{w,t} + c_w U_{w,t} \tag{10.1e}$$

A. Soroudi, *Power System Optimization Modeling in GAMS*,
DOI 10.1007/978-3-319-62350-4_10



Fig. 10.1 Water-energy nexus concept

Table 10.1 Technical and economical characteristics of thermal units

Unit	a_g($/MW^2)	b_g ($/MW)	c_g ($)	P_g^{max} (MW)	P_g^{min} (MW)
g_1	0.0002069	−0.1483	57.11	500	0
g_2	0.0003232	−0.1854	57.11	400	0
g_3	0.001065	−0.6026	126.8	400	0
g_4	0.0004222	−0.2119	57.11	350	0

$$P_g^{min} U_{g,t} \leq P_{g,t} \leq P_g^{max} U_{g,t} \tag{10.1f}$$

$$P_c^{min} U_{c,t} \leq P_{c,t} \leq P_c^{max} U_{c,t} \tag{10.1g}$$

$$W_c^{min} U_{c,t} \leq W_{c,t} \leq W_c^{max} U_{c,t} \tag{10.1h}$$

$$W_w^{min} U_{w,t} \leq W_{w,t} \leq W_w^{max} U_{w,t} \tag{10.1i}$$

$$R_c^{min} \leq \frac{P_{c,t}}{W_{c,t}} \leq R_c^{max} \tag{10.1j}$$

$$\sum_g P_{g,t} + \sum_c P_{c,t} = PL_t \tag{10.1k}$$

$$\sum_w W_{w,t} + \sum_c W_{c,t} = WL_t \tag{10.1l}$$

The simulation data are taken from [1] with slight modifications (Tables 10.1, 10.2, and 10.3).

Table 10.2 Technical and economical characteristics of co-production units

Unit	P_c^{max}	P_c^{min}	W_c^{max}	W_c^{min}	R_c^{min}	R_c^{max}	α_c	β_c	γ_c	ζ_c	ς_c	ξ_c
c_1	800	160	200	30	4	9	0.0004433	0.003546	0.007093	−1.106	−4.426	737.4
c_2	600	120	150	23	4	9	0.0007881	0.006305	0.01261	−1.475	−5.901	737.4
c_3	400	80	100	15	4	9	0.001773	0.01419	0.02837	−2.213	−8.851	737.4

Table 10.3 Technical and economical characteristics of water desalination units

Unit	a_w	b_w	c_w	W_w^{max}	W_w^{min}
w_1	0.00182	-0.708	7.374	250	0

GCode 10.1 Water-energy nexus optimization problem

```
Sets     t /t1*t24/,i /p1*p4/, c /c1*c3/, w /w1/;
Table gendata(i,*) generator cost characteristics and limits
   a          b        c       Pmax    Pmin
p1 0.0002069  −0.1483  57.11   500     0
p2 0.0003232  −0.1854  57.11   400     0
p3 0.001065   −0.6026  126.8   400     0
p4 0.0004222  −0.2119  57.11   350     0;
Table Coproduct(c,*)
     Pmax Pmin Wmax Wmin rmin rmax A11        A12       A22       b1      b2      C
c1   800  160  200  30   4    9    0.0004433  0.003546  0.007093  −1.106  −4.426  737.4
c2   600  120  150  23   4    9    0.0007881  0.006305  0.01261   −1.475  −5.901  737.4
c3   400  80   100  15   4    9    0.001773   0.01419   0.02837   −2.213  −8.851
     737.4;
Table waterdata(w,*)
     a          b          c      Wmax   Wmin
w1   1.82E−02  −7.081e−1  7.374   250    0;
Table PWdata(t,*)
     Pd    water
t1   1250  150
t2   1125  130
t3   875   100
t4   750   150
t5   950   200
t6   1440  350
t7   1500  300
t8   1750  200
t9   2000  300
t10  2250  400
t16  2500  550
t17  2125  550
t18  2375  500
t19  2250  400
t20  1975  350
t21  1750  300
t22  1625  250
t23  1500  200
t24  1376  150;
Variables  Of, p(i,t),TC,CC,Pc(c,t),Wc(c,t),Water(w,t),WaterCost;
Binary variables Up(i,t),Uc(c,t),Uw(w,t);
p.up(i,t)=gendata(i,"Pmax"); p.lo(i,t)=0; Pc.up(c,t)=Coproduct(c,'Pmax');
Pc.lo(c,t)=0; Wc.up(c,t)=Coproduct(c,'Wmax'); Wc.lo(c,t)=0;
Water.up(w,t)=waterdata(w,'Wmax'); Water.lo(w,t)=0;
Equations costThermal,balanceP,balanceW,costCoprodcalc,Objective,
costwatercalc,ratio1,ratio2,EQ1,EQ2,EQ3,EQ4,EQ5,EQ6,EQ7,EQ8;
costThermal..TC=e=sum((t,i), gendata(i,'a')*power(p(i,t),2)+gendata(i,'b')*p(i,t)
```

```
+gendata(i,'c')*Up(i,t));
balanceP(t) .. sum(i,p(i,t))+sum(c,Pc(c,t))=e=PWdata(t,'Pd');
balanceW(t) .. sum(w,Water(w,t))+sum(c,Wc(c,t))=e=PWdata(t,'water');
costCoprodcalc .. CC=e=sum((c,t),Coproduct(c,'A11')*power(Pc(c,t),2)
+2*Coproduct(c,'A12')*Pc(c,t)*Wc(c,t) +Coproduct(c,'A22')*power(Wc(c,t),2)
+Coproduct(c,'B1')*Pc(c,t)+Coproduct(c,'B2')*Wc(c,t)+Coproduct(c,'C')*Uc(c,t));
costwatercalc .. WaterCost=e=sum((t,w), waterdata(w,'a')*power(Water(w,t),2)
+waterdata(w,'b')*Water(w,t) +waterdata(w,'c')*Uw(w,t));
Objective .. OF=e=TC+CC+WaterCost;
ratio1(c,t) .. Pc(c,t)=l=Wc(c,t)*Coproduct(c,'Rmax');
ratio2(c,t) .. Pc(c,t)=g=Wc(c,t)*Coproduct(c,'Rmin');
eq1(w,t)     .. Water(w,t)=l=Uw(w,t)*waterdata(w,'Wmax');
eq2(w,t).. Water(w,t)=g=Uw(w,t)*waterdata(w,'Wmin');
eq3(c,t).. wc(c,t)=l= Uc(c,t)*Coproduct(c,'Wmax');
eq4(c,t).. wc(c,t)=g= Uc(c,t)*Coproduct(c,'Wmin');
eq5(c,t).. Pc(c,t)=l= Uc(c,t)*Coproduct(c,'Pmax');
eq6(c,t).. Pc(c,t)=g= Uc(c,t)*Coproduct(c,'Pmin');
eq7(i,t).. p(i,t) =l=Up(i,t)*gendata(i,"Pmax");
eq8(i,t).. p(i,t) =g=Up(i,t)*gendata(i,"Pmin");
Model DEDcostbased /all/; Solve DEDcostbased us MInlp min OF;
```

The hourly water output of different plants in water-energy nexus problem is depicted in Fig. 10.2. The hourly power output of different plants in water-energy nexus problem is shown in Fig. 10.3.



Fig. 10.2 Hourly water output of different plants in water-energy nexus problem



Fig. 10.3 Hourly power output of different plants in water-energy nexus problem



Fig. 10.4 Gas network linkage with electricity network

10.2 Gas-Power Nexus

The interaction of gas network and electricity network is modeled in this section. The electrical network is IEEE RTS 24-bus network which is shown in Fig. 10.4. It is a transmission network with the voltage levels of 138 kV, 230 kV and Sbase = 100 MVA. The branch data for IEEE RTS 24-bus network is given in Table 10.4 [2]. The from bus, to bus, reactance (X), resistance (r), total line charging susceptance (b), and MVA rating (MVA) are specified in this table. The parallel lines in MATPOWER are merged and the resultants are given in Table 10.4. The generation data for IEEE RTS 24-bus network is given in Table 10.5. The data of generating units in this network is inspired by Conejo et al. [3] and Bouffard et al. [4] with some modifications. The slack bus is bus 13 in this network. The wind turbines and the capacities are also shown in Fig. 10.4.

Table 10.4 Branch data for IEEE RTS 24-bus network

From	To	r(pu)	x(pu)	b(pu)	Rating (MVA)	From	To	r(pu)	x(pu)	b(pu)	Rating (MVA)
1	2	0.0026	0.0139	0.4611	175	11	13	0.0061	0.0476	0.0999	500
1	3	0.0546	0.2112	0.0572	175	11	14	0.0054	0.0418	0.0879	500
1	5	0.0218	0.0845	0.0229	175	12	13	0.0061	0.0476	0.0999	500
2	4	0.0328	0.1267	0.0343	175	12	23	0.0124	0.0966	0.2030	500
2	6	0.0497	0.1920	0.0520	175	13	23	0.0111	0.0865	0.1818	500
3	9	0.0308	0.1190	0.0322	175	14	16	0.0050	0.0389	0.0818	500
3	24	0.0023	0.0839	0.0000	400	15	16	0.0022	0.0173	0.0364	500
4	9	0.0268	0.1037	0.0281	175	15	21	0.0032	0.0245	0.2060	1000
5	10	0.0228	0.0883	0.0239	175	15	24	0.0067	0.0519	0.1091	500
6	10	0.0139	0.0605	2.4590	175	16	17	0.0033	0.0259	0.0545	500
7	8	0.0159	0.0614	0.0166	175	16	19	0.0030	0.0231	0.0485	500
8	9	0.0427	0.1651	0.0447	175	17	18	0.0018	0.0144	0.0303	500
8	10	0.0427	0.1651	0.0447	175	17	22	0.0135	0.1053	0.2212	500
9	11	0.0023	0.0839	0.0000	400	18	21	0.0017	0.0130	0.1090	1000
9	12	0.0023	0.0839	0.0000	400	19	20	0.0026	0.0198	0.1666	1000
10	11	0.0023	0.0839	0.0000	400	20	23	0.0014	0.0108	0.0910	1000
10	12	0.0023	0.0839	0.0000	400	21	22	0.0087	0.0678	0.1424	500

The gas network is also shown in Fig. 10.4 which its data is taken from [5]. The technical and economical characteristics of gas nodes are given in Table 10.6. The technical characteristics of gas network are also provided in Table 10.7 [5]. The gas network equations are described in (10.2).

$$\mathrm{GC} = \sum_{n,t} c_n \mathrm{Sg}_{n,t} \tag{10.2a}$$

$$\sum_{m} f_{n,m,t} = \sum_{m} f_{m,n,t} + \mathrm{Sg}_{n,t} - \zeta_{g,t}\mathrm{Sd}_n - Se_{n,t} \tag{10.2b}$$

$$f_{m,n,t} = C_{m,n}\sqrt{\mathrm{Pr}_{m,t}^2 - \mathrm{Pr}_{n,t}^2} \quad \text{Passive arcs} \tag{10.2c}$$

$$f_{m,n,t} \geq C_{mn}\sqrt{\mathrm{Pr}_{m,t}^2 - \mathrm{Pr}_{n,t}^2} \quad \text{Active arcs} \tag{10.2d}$$

$$\mathrm{Sg}_n^{\min} \leq \mathrm{Sg}_{n,t} \leq \mathrm{Sg}_n^{\max} \tag{10.2e}$$

$$\mathrm{Pr}_n^{\min} \leq \mathrm{Pr}_{n,t} \leq \mathrm{Pr}_n^{\max} \tag{10.2f}$$

Table 10.5 Generation data for IEEE RTS 24-bus network

Gen	Bus	$P_i^{\max}$	$P_i^{\min}$	b_i (\$/MW)	Cs_i (\$)	Cd_i (\$)	RU_i (MW h^{-1})	RD_i (MW h^{-1})	SU_i (MW h^{-1})	SD_i (MW h^{-1})	UT_i(h)	DT_i(h)	$u_{i,t=0}$	U_i^0 (h)	S_i^0 (h)
g1	18	400	100	5.47	0	0	47	47	105	108	1	1	1	5	0
g2	21	400	100	5.47	0	0	47	47	106	112	1	1	1	6	0
g3	1	152	30.4	13.32	1430.4	1430.4	14	14	43	45	8	4	1	2	0
g4	2	152	30.4	13.32	1430.4	1430.4	14	14	44	57	8	4	1	2	0
g5	15	155	54.25	16	0	0	21	21	65	77	8	8	0	0	2
g6	16	155	54.25	10.52	312	312	21	21	66	73	8	8	1	10	0
g7	23	310	108.5	10.52	624	624	21	21	112	125	8	8	1	10	0
g8	23	350	140	10.89	2298	2298	28	28	154	162	8	8	1	5	0
g9	7	350	75	20.7	1725	1725	49	49	77	80	8	8	0	0	2
g10	13	591	206.85	20.93	3056.7	3056.7	21	21	213	228	12	10	0	0	8
g11	15	60	12	26.11	437	437	7	7	19	31	4	2	0	0	1
g12	22	300	0	0	0	0	35	35	315	326	0	0	1	2	0

Table 10.6 Technical and economical characteristics of gas nodes

Gas node	Sg_n^{min} (10^6 Scm)	Sg_n^{max} (10^6 Scm)	Sd_n (10^6 Scm)	Pr_n^{min} (bar)	Pr_n^{max} (bar)	c_n $/MBtu
Anderlues	0	1.20	0.00	0.00	66.20	0.00
Antwerpen	0	0.00	4.03	1.25	80.00	0.00
Arlon	0	0.00	0.22	0.00	66.20	0.00
Berneau	0	0.00	0.00	0.00	66.20	0.00
Blaregnies	0	0.00	15.62	2.08	66.20	0.00
Brugge	0	0.00	3.92	1.25	80.00	0.00
Dudzele	0	8.40	0.00	0.00	77.00	2.28
Gent	0	0.00	5.26	1.25	80.00	0.00
Liege	0	0.00	6.39	1.25	66.20	0.00
Loenhout	0	4.80	0.00	0.00	77.00	2.28
Mons	0	0.00	6.85	0.00	66.20	0.00
Namur	0	0.00	2.12	0.00	66.20	0.00
Petange	0	0.00	1.92	1.04	66.20	0.00
Peronnes	0	0.96	0.00	0.00	66.20	1.68
Sinsin	0	0.00	0.00	0.00	63.00	0.00
Voeren	0	22.01	0.00	2.08	66.20	1.68
Wanze	0	0.00	0.00	0.00	66.20	0.00
Warnand	0	0.00	0.00	0.00	66.20	0.00
Zeebrugge	0	11.59	0.00	0.00	77.00	2.28
Zomergem	0	0.00	0.00	0.00	80.00	0.00

The electrical network equations are described in (10.3).

$$EC = \sum_{g,t} a_g(P_{g,t})^2 + b_g P_{g,t} + c_g + \sum_{i,t} \text{VOLL} \times \text{LS}_{i,t} + \text{VWC} \times P_{i,t}^{\text{wc}} \tag{10.3a}$$

$$\sum_{g \in \Omega_G^i} P_{g,t} + \text{LS}_{i,t} + P_{i,t}^{w} - L_{i,t} = \sum_{j \in \Omega_\ell^i} P_{ij,t} : \lambda_{i,t} \tag{10.3b}$$

$$P_{ij,t} = \frac{\delta_{i,t} - \delta_{j,t}}{X_{ij}} \tag{10.3c}$$

$$-P_{ij}^{\max} \leq P_{ij,t} \leq P_{ij}^{\max} \tag{10.3d}$$

$$P_g^{\min} \leq P_{g,t} \leq P_g^{\max} \tag{10.3e}$$

$$P_{g,t} - P_{g,t-1} \leq \text{RU}_g \tag{10.3f}$$

$$P_{g,t-1} - P_{g,t} \leq \text{RD}_g \tag{10.3g}$$

$$0 \leq \text{LS}_{i,t} \leq L_{i,t} \tag{10.3h}$$

Table 10.7 Technical characteristics of gas network

Pipe	From	To	Active	$C^2_{m,n}$
L_1	Zeebrugge	Dudzele		9.07027
L_2	Zeebrugge	Dudzele		9.07027
L_3	Dudzele	Brugge		6.04685
L_4	Dudzele	Brugge		6.04685
L_5	Brugge	Zomergem		1.39543
L_6	Loenhout	Antwerpen		0.10025
L_7	Antwerpen	Gent		0.14865
L_8	Gent	Zomergem		0.22689
L_9	Zomergem	Peronnes		0.65965
L_{10}	Voeren	Berneau	1	7.25622
L_{11}	Voeren	Berneau	1	0.10803
L_{12}	Berneau	Liege		1.81405
L_{13}	Berneau	Liege		0.02701
L_{14}	Liege	Warnand		1.45124
L_{15}	Liege	Warnand		0.02161
L_{16}	Warnand	Namur		0.86384
L_{17}	Namur	Anderlues		0.90703
L_{18}	Anderlues	Peronnes		7.25622
L_{19}	Peronnes	Mons		3.62811
L_{20}	Mons	Blaregnies		1.45124
L_{21}	Warnand	Wanze		0.05144
L_{22}	Wanze	Sinsin	1	0.00642
L_{23}	Sinsin	Arlon		0.00170
L_{24}	Arlon	Petange		0.02782

$$P^{wc}_{i,t} = w_t \Lambda^w_i - P^w_{i,t} \tag{10.3i}$$

$$0 \leq P^w_{i,t} \leq w_t \Lambda^w_i \tag{10.3j}$$

The overall optimization problem, constraints, and the decision variables are as follows:

$$\min_{\text{DV}} \text{OF} = \text{EC} + \text{GC} \tag{10.4}$$

$$\text{DV} = \left\{ \begin{array}{c} \delta_{i,t}, P_{g,t}, P_{c,t}, W_{c,t}, W_{w,t} \\ \text{Sg}_{n,t}, f_{n,m,t}, \text{Pr}_{n,t} \end{array} \right\}$$

(10.2 and (10.3)

The hourly variation pattern of wind generation, electric and gas demand is shown in Fig. 10.5.



Fig. 10.5 Hourly variation pattern of wind generation, electric and gas demand (pu)

The integrated electricity-gas problem is solved using the GCode 10.2. The total electricity cost EC is $\$3.9760 \times 10^5$. The total gas extraction costs are GC = $\$5.1755 \times 10^5$ and the total costs are $\$9.1515 \times 10^5$. The hourly variation pattern of wind generation, electric and gas demand is shown in Fig. 10.5.

GCode 10.2 Gas-electricity nexus optimization problem

```
Sets    bus  /1*24/  ,slack(bus) /13/,Gen /g1*g12/, t /t1*t24/
        genD(gen) /g1*g2,g5,g7*g11/, genN(gen) /g3,g4,g6,g12/ ;
scalars Sbase /100/   ,VOLL /10000/,VOLW /50/; alias(bus,node);
Sets gn NODES / Anderlues, Antwerpen, Arlon, Berneau, Blaregnies, Brugge, Dudzele,
                Gent, Liege, Loenhout, Mons, Namur, Petange, Peronnes, Sinsin,
                Voeren, Wanze, Warnand, Zeebrugge, Zomergem /
      a PIPES   / L1*L24 /; Alias (gn,gm); set Pnm(a,gn,gm) arc description;
table Ndata(gn,*) Node Data
            slo     sup       Sd      plo   pup   c
* Removed for saving space   ;
set GElink(gn,gen)
/Loenhout       .         g12
Voeren          .         g6
Sinsin          .         g3
Petange         .         g4/;
table AData(a,gn,gm,*) Arc Data
                                        act   C2mn;
table GD(Gen,*)   Generating units characteristics;
set GB(bus, Gen) connectivity index of each generating unit to each bus ;
Table BusData(bus,*) Demands of each bus in MW;
Table branch(bus, node,*)     Network technical
                r        x        b      z   limit;
Table DataWDL(t,*)
     w                   d                     g;
Parameters Wcap(bus),conex(bus,node),SD(gn);
branch(bus,node,'bij')$branch(bus,node,'Limit') =1/branch(bus,node,'x');
conex(bus,node)$(branch(bus,node,'limit')and branch(node,bus,'limit'))=1;
conex(bus,node)$(conex(node,bus))=1; Variables f(a,gn,gm,t),sg(gn,t),pressure(gn,t)
     ,
EC,Pij(bus,node,t),Pg(Gen,t),delta(bus,t),lsh(bus,t),Pw(bus,t),pc(bus,t),Gc,OF ;
Pnm(a,gn,gm)$adata(a,gn,gm,'c2mn')=yes;
Equations const1,const2,const3,const4,const5,const6,CG1,CG2,CG3,CG4,Objective;
const1(bus,node,t)$conex(bus,node).. Pij(bus,node,t)=e=
```

```
branch(bus,node,'bij')*(delta(bus,t)-delta(node,t));
const2(bus,t)..lsh(bus,t)$BusData(bus,'pd')+Pw(bus,t)$Wcap(bus)+sum(Gen$GB(bus,Gen)
      ,
Pg(Gen,t))-DataWDL(t,'d')*BusData(bus,'pd')/Sbase=e=
+sum(node$conex(node,bus),Pij(bus,node,t));
const3..EC=e=sum((bus,GenD,t)$GB(bus,GenD),Pg(GenD,t)*GD(GenD,'b')*Sbase)
+sum((bus,t),VOLL*lsh(bus,t)*Sbase$BusData(bus,'pd')+VOLW*Pc(bus,t)*sbase$Wcap(bus)
      );
const4(gen,t) .. pg(gen,t+1)-pg(gen,t)=l=GD(gen,'RU')/Sbase;
const5(gen,t) .. pg(gen,t-1)-pg(gen,t)=l=GD(gen,'RD')/Sbase;
const6(bus,t)$Wcap(bus) .. pc(bus,t)=e=DataWDL(t,'w')*Wcap(bus)/Sbase-pw(bus,t);
Pg.lo(Gen,t)=GD(Gen,'Pmin')/Sbase; Pg.up(Gen,t)=GD(Gen,'Pmax')/Sbase;
delta.up(bus,t)=pi/2; delta.lo(bus,t)=-pi/2; delta.fx(slack,t)=0;
Pij.up(bus,node,t)$((conex(bus,node)))=1* branch(bus,node,'Limit')/Sbase;
Pij.lo(bus,node,t)$((conex(bus,node)))=-1*branch(bus,node,'Limit')/Sbase;
lsh.up(bus,t)= DataWDL(t,'d')*BusData(bus,'pd')/Sbase; lsh.lo(bus,t)= 0;
Pw.up(bus,t)=DataWDL(t,'w')*Wcap(bus)/Sbase; Pw.lo(bus,t)=0;
Pc.up(bus,t)=DataWDL(t,'w')*Wcap(bus)/Sbase; Pc.lo(bus,t)=0; SD(gn)=Ndata(gn,'SD');
CG1(gn,t).. sum(Pnm(a,gn,gm),f(Pnm,t))=e=sum(Pnm(a,gm,gn),f(Pnm,t))
+sg(gn,t)$(Ndata(gn,'Sup')>0)-DataWDL(t,'G')*SD(gn)
-sum((GenN)$Gelink(gn,GenN),Pg(GenN,t)*GD(GenN,'b')*Sbase/35315);
CG2(Pnm(a,gn,gm),t)$(AData(a,gn,gm,'C2mn') AND AData(a,gn,gm,'ACT')=0)
  .. signpower(f(Pnm,t),2) =e= AData(Pnm,'C2mn')*(pressure(gn,t)-pressure(gm,t));
CG3(Pnm(a,gn,gm),t)$(AData(a,gn,gm,'C2mn') AND AData(a,gn,gm,'ACT')=1)
   .. -sqr(f(Pnm,t)) =g= AData(Pnm,'C2mn')*(pressure(gn,t)-pressure(gm,t));
CG4.. Gc =e= sum((gn,t), 35315*ndata(gn,'c')*sg(gn,t)$Ndata(gn,'Sup'));
Objective..OF=e=EC+Gc;sg.lo(gn,t)=0; sg.up(gn,t)= ndata(gn,'sup');
pressure.lo(gn,t) = sqr(ndata(gn,'plo')); pressure.up(gn,t) = sqr(ndata(gn,'pup'));
f.lo(Pnm(a,gn,gm),t)$(AData(a,gn,gm,'C2mn')) =
-sqrt(AData(a,gn,gm,'C2mn')*( pressure.up(gn,t)-pressure.lo(gn,t)));
f.up(Pnm(a,gn,gm),t)$(AData(a,gn,gm,'C2mn'))=
sqrt(AData(a,gn,gm,'C2mn')*( pressure.up(gn,t)-pressure.lo(gn,t)));
f.lo(Pnm(a,gn,gm),t)$(AData(a,gn,gm,'C2mn') AND AData(a,gn,gm,'ACT')=1) =0;
Model overall /all/; Solve overall using nlp min OF;
```



Fig. 10.6 Hourly variation pattern of gas generation from gas sources

Hourly variation pattern of gas generation from gas sources are shown in Fig. 10.6.

The hourly variation pattern of electricity power generation is shown in Fig. 10.7.



Fig. 10.7 Hourly variation pattern of electricity power generation

The interaction of gas network and electricity sector has been investigated in several works. The impacts of the gas network on security constrained UC is analyzed in [6]. A unified gas and power flow analysis in natural gas and electricity coupled networks can be found in [7]. A robust scheduling model for wind-integrated energy systems with the considerations of both gas pipeline and power transmission contingencies is developed in [8]. The reliability of gas networks and their impacts on the reliability of electricity network is modeled in [9]. The impact of large penetration of wind generation on the UK gas network is analyzed in [10]. One of the recent efficient methods of electricity storage is storing the electricity as gas. This method is also called power to gas or P2G technique [11].

10.3 Energy Hub Concept

The concept of Energy Hub was introduced in [12]. Energy hub may be considered as a virtual box that can convert a set of input energy carriers into a set of energy demands. This box contains several technologies that can store, transfer, or convert different forms of energies to each other. A general example of Energy hub is shown in Fig. 10.8. Different aspects of energy hubs are investigated in the literature such as economic dispatch of energy hubs [13], demand response and energy hub [14], energy hub concept applied on car manufacturing plants [15], and wind power uncertainty modeling in energy hubs [16, 17].

The technologies shown in Fig. 10.8 are explained as follows:

- Combined heat and power (CHP): receives the natural gas (G_t) and converts it into heat (H_t) and electricity (E_t) The CHP economic dispatch problem can be modeled in (10.5) [18]:

$$H_t = \eta_{\text{ge}}^{\text{chp}} G_t \tag{10.5a}$$



Fig. 10.8 The energy hub concept

$$E_t = \eta_{gh}^{chp} G_t \tag{10.5b}$$

- Electric heat pump (EHP): It is fed by electricity and generates heat demand (H_t) or cool demand (C_t) based on the operating mode. The operation of EHP is mathematically formulated as follows:

$$C_t + H_t = E_t \times \text{COP} \tag{10.6a}$$

$$H_t^{\min} I_t^h \le H_t \le H_t^{\max} I_t^h \tag{10.6b}$$

$$C_t^{\min} I_t^c \le C_t \le C_t^{\max} I_t^c \tag{10.6c}$$

$$I_t^c + I_t^h \le 1 \tag{10.6d}$$

$$I_t^c, I_t^h \in \{0, 1\}$$

The EHP can be in heat or cool generation mode. COP is the coefficient of performance for EHP.

- Chiller boiler (CB): It receives heat and transforms it into cool demand The chiller boiler operation is mathematically formulated as follows:

$$C_t = \eta_{hc} H_t \tag{10.7}$$

The η_{hc} is the efficiency of heat to cooling conversion for chiller boiler.

- Electricity storage system (ESS): It can store (electricity) and then discharge electricity The ESS operation is mathematically formulated as follows:

$$\mathrm{SOC}_t = \mathrm{SOC}_{t-1} + (E_t^{\mathrm{ch}}\eta_c - E_t^{\mathrm{dch}}/\eta_d)\Delta_t \tag{10.8a}$$

$$E_{\min}^{\mathrm{ch}} \le E_t^{\mathrm{ch}} \le E_{\max}^{\mathrm{dch}} \tag{10.8b}$$

$$E_{\min}^{\mathrm{dch}} \le E_t^{\mathrm{dch}} \le E_{\max}^{\mathrm{dch}} \tag{10.8c}$$

$$\mathrm{SOC}_{\min} \le \mathrm{SOC}_t \le \mathrm{SOC}_{\max} \tag{10.8d}$$

$$I_t^{\mathrm{dch}} + I_t^{\mathrm{ch}} \le 1 \tag{10.8e}$$

$$I_t^{\mathrm{ch}}, I_t^{\mathrm{dch}} \in \{0, 1\}$$

  SOC_t is the state of charge in ESS. $E_t^{\mathrm{ch/dch}}$ is for demonstrating the charged and discharged electricity in ESS. The binary variables $I_t^{\mathrm{ch}}, I_t^{\mathrm{dch}}$ show the charge or discharge mode of ESS at time t.
- Transformer (Tr): It receives electricity and the output is also electricity (with different voltage level)

$$E_t^{\mathrm{out}} = \eta_{\mathrm{ee}} E_t^{\mathrm{in}} \tag{10.9}$$

- Furnace (F): receives the natural gas and generates the heat demand

$$H_t = \eta_{\mathrm{gh}} G_t \tag{10.10}$$

Three different energy hub configurations will be analyzed to investigate the level of achievable operational flexibility.

10.3.1 Data

The energy hubs which are analyzed in this chapter would have three types of energy demands namely electric, heat, and cooling demand. Different hourly demand and electricity price data for three energy hub configurations are given in Table 10.8.

- The charging and discharging efficiencies ($\eta_{\mathrm{ch/dch}}$) of ESS are assumed to be 0.9. The ESS capacity is $\mathrm{SOC}_{\max} = 600$ MW h and $\mathrm{SOC}_{\min} = 120$ MW h. The initial stored energy in ESS is 120 MW h. The minimum charging and discharging limits are $E_{\min}^{\mathrm{ch/dch}} = 0$ and the maximum charging and discharging limits are $E_{\max}^{\mathrm{ch/dch}} = 120$ MW.
- The transformer efficiency is $\eta_{\mathrm{ee}} = 0.98$.
- The CHP efficiencies for gas to electricity is $\eta_{\mathrm{ge}} = 0.35$ and for gas to heat is $\eta_{\mathrm{gh}} = 0.45$. The CHP capacity is 250 MW.

Table 10.8 Different hourly demand and electricity price data for energy hub configurations

Time	D_t^h (MW)	D_t^e (MW)	D_t^c (MW)	λ_t^e \$/MW h
t_1	21.41	52.10	11.51	36.67
t_2	23.21	66.70	13.68	40.41
t_3	26.09	72.20	16.01	38.48
t_4	26.72	78.37	21.42	38.00
t_5	25.59	120.20	21.97	40.24
t_6	26.45	83.48	30.80	38.55
t_7	39.54	110.40	38.94	52.26
t_8	47.28	124.29	46.78	67.34
t_9	52.12	143.61	50.97	70.47
t_{10}	49.13	149.28	48.86	66.20
t_{11}	69.26	154.19	34.77	73.30
t_{12}	61.97	147.30	32.68	60.82
t_{13}	68.04	200.71	27.77	63.15
t_{14}	68.56	174.37	32.02	70.77
t_{15}	56.40	176.54	33.22	63.09
t_{16}	41.32	136.11	34.13	52.53
t_{17}	37.43	108.71	40.78	57.00
t_{18}	25.44	96.90	43.56	49.15
t_{19}	25.66	89.08	51.48	47.47
t_{20}	21.94	82.49	43.15	49.46
t_{21}	22.44	76.93	36.49	53.07
t_{22}	24.63	66.85	27.68	51.60
t_{23}	22.72	47.17	19.14	50.53
t_{24}	22.59	64.67	11.04	36.38

- For EHP, $C_{\text{Max}}^{\text{ehp}} = H_{\text{Max}}^{\text{ehp}} = 500\,\text{MW}$; The COP is assumed to be 2.5.
- The furnace efficiency η_{gh}^{f} is 0.9. The furnace capacity is 600 MW.
- The chiller boiler has a capacity equal to 500 MW and the efficiency is $\eta_{\text{hc}} = 0.95$
- The natural gas price is assumed to be constant for different hours and it is equal to $\lambda_t^g = 12\$/\text{MW h}$

10.3.2 Configuration I

This configuration contains transformer, furnace, and chiller boiler as shown in Fig. 10.9.

$$\min \text{OF} = \sum_t \lambda_t^e E_t + \lambda_t^g G_t \tag{10.11a}$$

$$\eta_{\text{ee}} E_t = D_t^e \tag{10.11b}$$



Fig. 10.9 The energy hub configuration I, considering transformer, furnace, and chiller boiler

$$\eta_{\mathrm{gh}}^{f} G_t = H_{1,t} + H_{2,t} \tag{10.11c}$$

$$\eta_{\mathrm{hc}} H_{2,t} = D_t^c \tag{10.11d}$$

$$H_{1,t} = D_t^h \tag{10.11e}$$

The GAMS code for solving the hub configuration I described in (10.11) is given in GCode 10.3.

The developed model in GCode 10.3 is linear and can be solved using any lp solver. The problem is solved and the total operating costs are $\$1.1327 \times 10^5$. The hourly purchased electricity and natural gas in energy hub configuration I are shown in Fig. 10.10.

The output of furnace system will be divided into two streams. The first one will supply the chiller and the second one will directly supply the heat demand. The hourly output of furnace unit in energy hub configuration-I is shown in Fig. 10.11.

GCode 10.3 The optimal operation of energy hub configuration I

```
Set      t       hours          /t1*t24/
Table  data(t,*)
      Dh    De     Dc    lambda
t1    21.4  52.1   11.5  36.7
t2    23.2  66.7   13.7  40.4
t3    26.1  72.2   16    38.5
t4    26.7  78.4   21.4  38
t5    25.6  120.2  22    40.2
t6    26.4  83.5   30.8  38.6
t7    39.5  110.4  38.9  52.3
t8    47.3  124.3  46.8  67.3
t9    52.1  143.6  51    70.5
t10   49.1  149.3  48.9  66.2
t11   69.3  154.2  34.8  73.3
t12   62    147.3  32.7  60.8
t13   68    200.7  27.8  63.2
t14   68.6  174.4  32    70.8
t15   56.4  176.5  33.2  63.1
t16   41.3  136.1  34.1  52.5
t17   37.4  108.7  40.8  57
t18   25.4  96.9   43.6  49.2
t19   25.7  89.1   51.5  47.5
t20   21.9  82.5   43.1  49.5
t21   22.4  76.9   36.5  53.1
t22   24.6  66.8   27.7  51.6
t23   22.7  47.2   19.1  50.5
t24   22.6  64.7   11    36.4
* ---------------------------------------------------------------
Variable  cost;
Positive  variables    E(t),G(t),H1(t),H2(t);
Scalar
CBmax  /500/,  eta_ee  /0.98/,eta_ghf  /0.9/,eta_hc  /0.95/;
H2.up(t)=CBmax;
Equations
eq1,eq2,eq3,eq4,eq5;
eq1        ..    cost  =e=sum(t,data(t,'lambda')*E(t)+12*G(t));
eq2(t)     ..    eta_ee*E(t)=e=data(t,'E');
eq3(t)   ..    H1(t)=e=data(t,'h');
eq4(t)   ..    eta_ghf*G(t)=e=H1(t)+H2(t);
eq5(t)   ..    eta_hc*H2(t)=e=data(t,'c');
Model  Hub  /all/;
Solve  hub  us  lp  min  cost;
```

10.3.3 Configuration II

This configuration contains transformer, furnace, chiller boiler, CHP, and ESS as shown in Fig. 10.12.



Fig. 10.10 The hourly purchased electricity and natural gas in energy hub configuration I



Fig. 10.11 The hourly output of furnace unit in energy hub configuration I



Fig. 10.12 The energy hub configuration II, considering transformer, furnace, chiller boiler, CHP, and electric energy storage

$$\min \mathrm{OF} = \sum_t \lambda_t^e E_t + \lambda_t^g G_t \tag{10.12a}$$

$$\eta_{\mathrm{ee}} E_{2,t} + E_t^{\mathrm{dch}} + \eta_{\mathrm{ge}} G_{1,t} = D_t^e \tag{10.12b}$$

$$E_t = E_{1,t} + E_{2,t} \tag{10.12c}$$

$$E_{1,t} = E_t^{\mathrm{ch}} \tag{10.12d}$$

$$\mathrm{SOC}_t = \mathrm{SOC}_{t-1} + (E_t^{\mathrm{ch}} \eta_c - E_t^{\mathrm{dch}} / \eta_d) \Delta_t \tag{10.12e}$$

$$E_{\min}^{\mathrm{ch}} \leq E_t^{\mathrm{ch}} \leq E_{\max}^{\mathrm{dch}} \tag{10.12f}$$

$$E_{\min}^{\mathrm{dch}} \leq E_t^{\mathrm{dch}} \leq E_{\max}^{\mathrm{dch}} \tag{10.12g}$$

$$\mathrm{SOC}_{\min} \leq \mathrm{SOC}_t \leq \mathrm{SOC}_{\max} \tag{10.12h}$$

$$I_t^{\mathrm{dch}} + I_t^{\mathrm{ch}} \leq 1 \tag{10.12i}$$

$$I_t^{\mathrm{ch}}, I_t^{\mathrm{dch}} \in \{0, 1\}$$

$$G_t = G_{1,t} + G_{2,t} \tag{10.12j}$$

$$\eta_{\mathrm{gh}}^f G_{1,t} + H_{1,t} = D_t^h \tag{10.12k}$$

$$\eta_{gh} G_{2,t} = H_{1,t} + H_{2,t} \tag{10.12l}$$

$$\eta_{hc} H_{2,t} = D_t^c \tag{10.12m}$$

The GAMS code for solving the hub configuration II described in (10.12) is given in GCode 10.4.

GCode 10.4 The optimal operation of energy hub configuration II

```
set      t        hours            /t1*t24/

table  data(t,*)
      Dh     De      Dc    lambda
t1    21.4  52.1    11.5  36.7
t2    23.2  66.7    13.7  40.4
t3    26.1  72.2    16    38.5
t4    26.7  78.4    21.4  38
t5    25.6  120.2   22    40.2
t6    26.4  83.5    30.8  38.6
t7    39.5  110.4   38.9  52.3
t8    47.3  124.3   46.8  67.3
t9    52.1  143.6   51    70.5
t10   49.1  149.3   48.9  66.2
t11   69.3  154.2   34.8  73.3
t12   62    147.3   32.7  60.8
t13   68    200.7   27.8  63.2
t14   68.6  174.4   32    70.8
t15   56.4  176.5   33.2  63.1
t16   41.3  136.1   34.1  52.5
t17   37.4  108.7   40.8  57
t18   25.4  96.9    43.6  49.2
t19   25.7  89.1    51.5  47.5
t20   21.9  82.5    43.1  49.5
t21   22.4  76.9    36.5  53.1
t22   24.6  66.8    27.7  51.6
t23   22.7  47.2    19.1  50.5
t24   22.6  64.7    11    36.4
Variable cost;
Positive variables
E(t),E1(t),E2(t),G(t),G1(t),G2(t),H1(t),H2(t)
SOC(t),Ec(t),Ed(t) ;
Binary variables Idch(t),Ich(t);
scalar SOC0 /20/,SOCmax /600/,eta_c /0.9/,eta_d /0.9/,eta_ee /0.98/,eta_ge /0.45/,
eta_gh /0.35/,eta_hc /0.95/,Chpmax /250/,CBmax /500/,Fmax /600/,eta_ghf /0.9/;
SOC0= 0.2*SOCmax;
SOC.up(t)=SOCmax; SOC.lo(t)=0.2*SOCmax; SOC.fx('t24')=SOC0;
Ec.up(t)=0.2*SOCmax; Ec.lo(t)=0;
Ed.up(t)=0.2*SOCmax; Ed.lo(t)=0;
G1.up(t)=Chpmax;
G2.up(t)=Fmax;
H2.up(t)=CBmax;
Equations  eq1,eq2,eq3,eq4,eq5,eq6,eq7,eq8,eq9,eq10,eq11,eq12;
eq1        ..    cost =e=sum(t,data(t,'lambda')*E(t)+12*G(t));
eq2(t)     ..    eta_ee*E2(t)+Ed(t)+eta_ge*G1(t)=e=data(t,'E');
eq3(t)     ..    E(t)=e=E1(t)+E2(t);
eq4(t)     ..    E1(t)=e=Ec(t);
eq5(t)     ..    SOC(t)=e=SOC0$(ord(t)=1)+ SOC(t-1)$(ord(t)>1)+Ec(t)*eta_c-Ed(t)/eta_d
     ;
eq6(t)     ..    Ed(t)=l=0.2*SOCmax*Idch(t);
eq7(t)     ..    Ec(t)=l=0.2*SOCmax*Ich(t);
eq8(t)     ..    Idch(t)+Ich(t)=l=1;
```

```
eq9(t) ..  G(t)=e=G1(t)+G2(t);
eq10(t) ..  eta_gh*G1(t)+H1(t)=e=1*data(t,'h');
eq11(t) ..  eta_ghf*G2(t)=e=H1(t)+H2(t);
eq12(t) ..  eta_hc*H2(t)=e=data(t,'c');
model Hub2 /all/;
Solve hub2 us mip min cost;
```

The developed model in GCode 10.4 is linear and can be solved using any lp solver. The problem is solved, and the total operating costs are \$0.85504 × 10^5. The hourly purchased electricity and its division between transformer and ESS in energy hub configuration-II are shown in Fig. 10.13. The output of furnace system will be divided into two streams. The first one will supply the chiller and the second one will supply the heat demand. The hourly output of furnace unit in energy hub configuration II is shown in Fig. 10.14. The hourly state of charge, charging, and discharging of ESS in energy hub configuration II is shown in Fig. 10.15.

10.3.4 Configuration III

This configuration contains transformer, furnace, chiller boiler, CHP, and ESS as shown in Fig. 10.16.



Fig. 10.13 The hourly purchased electricity and its division between transformer and ESS in energy hub configuration II



Fig. 10.14 The hourly output of furnace unit in energy hub configuration II



Fig. 10.15 The hourly state of charge (MW h), charging, and discharging of ESS (MW) in energy hub configuration II

$$\min \text{OF} = \sum_t \lambda_t^e E_t + \lambda_t^g G_t \tag{10.13a}$$

$$\eta_{\text{ee}} E_{2,t} + E_t^{\text{dch}} + \eta_{\text{ge}} G_{1,t} = D_t^e \tag{10.13b}$$

$$E_t = E_{1,t} + E_{2,t} + E_{3,t} \tag{10.13c}$$

$$E_{1,t} = E_t^{\text{ch}} \tag{10.13d}$$



Fig. 10.16 The energy hub configuration III, considering transformer, furnace, chiller boiler, CHP, electric energy storage, and EHP

$$\mathrm{SOC}_t = \mathrm{SOC}_{t-1} + (E_t^{\mathrm{ch}}\eta_c - E_t^{\mathrm{dch}}/\eta_d)\Delta_t \tag{10.13e}$$

$$E_{\min}^{\mathrm{ch}} I_t^{\mathrm{ch}} \le E_t^{\mathrm{ch}} \le E_{\max}^{\mathrm{ch}} I_t^{\mathrm{ch}} \tag{10.13f}$$

$$E_{\min}^{\mathrm{dch}} I_t^{\mathrm{dch}} \le E_t^{\mathrm{dch}} \le E_{\max}^{\mathrm{dch}} I_t^{\mathrm{dch}} \tag{10.13g}$$

$$\mathrm{SOC}_{\min} \le \mathrm{SOC}_t \le \mathrm{SOC}_{\max} \tag{10.13h}$$

$$I_t^{\mathrm{dch}} + I_t^{\mathrm{ch}} \le 1 \tag{10.13i}$$

$$I_t^{\mathrm{ch}}, I_t^{\mathrm{dch}} \in \{0, 1\}$$

$$G_t = G_{1,t} + G_{2,t} \tag{10.13j}$$

$$\eta_{\mathrm{gh}} G_{1,t} + H_{1,t} + H_t^{\mathrm{EHP}} = D_t^h \tag{10.13k}$$

$$\eta_{\mathrm{gh}}^f G_{2,t} = H_{1,t} + H_{2,t} \tag{10.13l}$$

$$\eta_{\mathrm{hc}} H_{2,t} + C_t^{\mathrm{EHP}} = D_t^c \tag{10.13m}$$

$$C_t^{\mathrm{EHP}} + H_t^{\mathrm{EHP}} = E_{3,t} \times \mathrm{COP} \tag{10.13n}$$

$$H_t^{\min} I_t^h \le H_t^{\mathrm{EHP}} \le H_t^{\max} I_t^h \tag{10.13o}$$

$$C_t^{\min} I_t^c \le C_t^{\mathrm{EHP}} \le C_t^{\max} I_t^c \tag{10.13p}$$

$$I_t^c + I_t^h \leq 1 \tag{10.13q}$$

$$I_t^c, I_t^h \in \{0, 1\}$$

The developed model in GCode 10.5 is linear and can be solved using any lp solver.

GCode 10.5 The optimal operation of energy hub configuration III

```
Set      t        hours              /t1*t24/

Table  data(t,*)
         Dh     De     Dc     lambda
t1    21.4  52.1   11.5  36.7
t2    23.2  66.7   13.7  40.4
t3    26.1  72.2   16    38.5
t4    26.7  78.4   21.4  38
t5    25.6  120.2  22    40.2
t6    26.4  83.5   30.8  38.6
t7    39.5  110.4  38.9  52.3
t8    47.3  124.3  46.8  67.3
t9    52.1  143.6  51    70.5
t10   49.1  149.3  48.9  66.2
t11   69.3  154.2  34.8  73.3
t12   62    147.3  32.7  60.8
t13   68    200.7  27.8  63.2
t14   68.6  174.4  32    70.8
t15   56.4  176.5  33.2  63.1
t16   41.3  136.1  34.1  52.5
t17   37.4  108.7  40.8  57
t18   25.4  96.9   43.6  49.2
t19   25.7  89.1   51.5  47.5
t20   21.9  82.5   43.1  49.5
t21   22.4  76.9   36.5  53.1
t22   24.6  66.8   27.7  51.6
t23   22.7  47.2   19.1  50.5
t24   22.6  64.7   11    36.4   ;

data(t,'lambda')=0.6*data(t,'lambda');
variable  cost;
positive  variables  E(t),E1(t),E2(t),E3(t),G(t),G1(t),G2(t),H1(t),H2(t)
SOC(t),Ec(t),Ed(t),H_ehp(t),C_ehp(t)  ;
Binary  variables  Idch(t),Ich(t),Ic(t),Ih(t);
scalar  SOC0  /20/,  SOCmax  /600/,  eta_c  /0.9/,  eta_d  /0.9/,  eta_ee  /0.98/
,eta_ge  /0.45/  ,eta_gh  /0.35/
eta_hc  /0.95/,  COP  /2.5/,  H_ehpMax  /200/,  C_ehpMax  /200/,Chpmax  /300/,
CBmax  /300/,Fmax  /300/,eta_ghf  /0.9/;
SOC0=  0.2*SOCmax;  SOC.up(t)=SOCmax;  SOC.lo(t)=0.2*SOCmax;  SOC.fx('t24')=SOC0;
Ec.up(t)=0.2*SOCmax;  Ec.lo(t)=0;  Ed.up(t)=0.2*SOCmax;  Ed.lo(t)=0;
C_ehp.up(t)=C_ehpMax;  H_ehp.up(t)=H_ehpMax;
G1.up(t)=Chpmax;  G2.up(t)=Fmax;  H2.up(t)=CBmax;  E.up(t)=1000;
Equations
eq1,eq2,eq3,eq4,eq5,eq6,eq7,eq8,eq9,eq10,eq11,eq12,eq13,eq14,eq15,eq16;

eq1       ..    cost  =e=sum(t,data(t,'lambda')*E(t)+12*G(t));
eq2(t)    ..    eta_ee*E2(t)  +Ed(t)+eta_ge*G1(t)=e=data(t,'E')+E3(t);
eq3(t)    ..    E(t)=e=E1(t)+E2(t);
eq4(t)    ..    E1(t)=e=Ec(t);
eq5(t)    ..    SOC(t)=e=SOC0$(ord(t)=1)+ SOC(t-1)$(ord(t)>1)+Ec(t)*eta_c-Ed(t)/eta_d
      ;
eq6(t)    ..    Ed(t)=l=0.2*SOCmax*Idch(t);
```

```
eq7(t)    ..   Ec(t)=l=0.2*SOCmax*Ich(t);
eq8(t)    ..   Idch(t)+Ich(t)=l=1;
eq9(t)    ..   G(t)=e=G1(t)+G2(t);
eq10(t)   ..   eta_gh*G1(t)+H1(t)+H_ehp(t)=e=1*data(t,'h');
eq11(t)   ..   eta_ghf*G2(t)=e=H1(t)+H2(t);
eq12(t)   ..   eta_hc*H2(t)+C_ehp(t)=e=data(t,'c');
eq13(t)   ..   C_ehp(t)+H_ehp(t)=e=E3(t)*cop;
eq14(t)   ..   H_ehp(t)=l=H_ehpMax * Ih(t);
eq15(t)   ..   C_ehp(t)=l=C_ehpMax * Ic(t);
eq16(t)   ..   Ic(t)+Ih(t)=l=1;
Model Hub /all/;
Solve hub us mip min cost;
```



Fig. 10.17 The hourly purchased electricity and its division between transformer and ESS in energy hub configuration III

The problem is solved and the total operating costs are 0.84430×10^5. The hourly purchased electricity and its division between transformer and ESS in energy hub configuration-III are shown in Figs. 10.17, 10.18. The output of furnace system will be divided into two streams. The first one will supply the chiller and the second one will supply the heat demand. The hourly output of furnace unit in energy hub configuration II is shown in Fig. 10.14. The hourly state of charge, charging, and discharging of ESS in energy hub configuration III is shown in Fig. 10.19. The hourly output of EHP in energy hub configuration III is shown in Fig. 10.20.



Fig. 10.18 The hourly output of furnace unit in energy hub configuration III



Fig. 10.19 The hourly state of charge (MW h), charging, and discharging of ESS (MW) in energy hub configuration III



Fig. 10.20 The hourly output of EHP in energy hub configuration III

References

1. A. Santhosh, A.M. Farid, K. Youcef-Toumi, Real-time economic dispatch for the supply side of the energy-water nexus. Appl. Energy **122**, 42–52 (2014)
2. R.D. Zimmerman, C.E. Murillo-Sanchez, R.J. Thomas, Matpower: steady-state operations, planning, and analysis tools for power systems research and education. IEEE Trans. Power Syst. **26**(1), 12–19 (2011)
3. A.J. Conejo, M. Carrión, J.M. Morales, *Decision Making Under Uncertainty in Electricity Markets*, vol. 1 (Springer, Berlin, 2010)
4. F. Bouffard, F.D. Galiana, A.J. Conejo, Market-clearing with stochastic security-part II: case studies. IEEE Trans. Power Syst. **20**(4), 1827–1835 (2005)
5. D. De Wolf, Y. Smeers, The gas transmission problem solved by an extension of the simplex algorithm. Manag. Sci. **46**(11), 1454–1465 (2000)
6. T. Li, M. Eremia, M. Shahidehpour, Interdependency of natural gas network and power system security. IEEE Trans. Power Syst. **23**(4), 1817–1824 (2008)
7. A. Martinez-Mares, C.R. Fuerte-Esquivel, A unified gas and power flow analysis in natural gas and electricity coupled networks. IEEE Trans. Power Syst. **27**(4), 2156–2166 (2012)
8. L. Bai, F. Li, T. Jiang, H. Jia, Robust scheduling for wind integrated energy systems considering gas pipeline and power transmission n-1 contingencies. IEEE Trans. Power Syst. **32**(2), 1582–1584 (2017)
9. J. Munoz, N. Jimenez-Redondo, J. Perez-Ruiz, J. Barquin, Natural gas network modeling for power systems reliability studies. in *2003 IEEE Bologna Power Tech Conference Proceedings*, vol. 4, June 2003, p. 8
10. M. Qadrdan, M. Chaudry, J. Wu, N. Jenkins, J. Ekanayake, Impact of a large penetration of wind generation on the GB gas network. Energy Policy **38**(10), 5684–5695 (2010)

11. H.S. de Boer, L. Grond, H. Moll, R. Benders, The application of power-to-gas, pumped hydro storage and compressed air energy storage in an electricity system at different wind power penetration levels. Energy **72**, 360–370 (2014)
12. M. Geidl, G. Koeppel, P. Favre-Perrod, B. Klöckl, G. Andersson, K. Fröhlich, The energy hub–a powerful concept for future energy systems. in *Third Annual Carnegie Mellon Conference on the Electricity Industry*, Pittsburgh (2007), pp. 13–14
13. S.D. Beigvand, H. Abdi, M. La Scala, A general model for energy hub economic dispatch. Appl. Energy **190**, 1090–1111 (2017)
14. M. Batić, N. Tomašević, G. Beccuti, T. Demiray, S. Vraneš, Combined energy hub optimisation and demand side management for buildings. Energy Build. **127**, 229–241 (2016)
15. K. Kampouropoulos, F. Andrade, Energy hub optimization applied on car manufacturing plants. in *ANDESCON, 2016 IEEE* (IEEE, Piscataway, 2016), pp. 1–4
16. A. Soroudi, B. Mohammadi-Ivatloo, A. Rabiee, Energy hub management with intermittent wind power. in *Large Scale Renewable Power Generation* (Springer, Singapore, 2014), pp. 413–438
17. A. Dolatabadi, B. Mohammadi-Ivatloo, M. Abapour, S. Tohidi, Optimal stochastic design of wind integrated energy hub. IEEE Trans. Ind. Inf. **99**, 1–1 (2017). doi:10.1109/TII.2017.2664101
18. M. Geidl, Integrated modeling and optimization of multi-carrier energy systems. PhD thesis, TU Graz, 2007